Добрый день !
ERP 2
Подскажите в чем проблема в следующей ситуации: в ФормуДокумента есть кнопка авкОСВ - формирующая отчет ОСВ;
добавил свою кнопку авкОСВ1, - для формирования ОСВ по счету 62, которая вызывает авкОСВ
Проблема: если вначале нажать авкОСВ - сформируется ОСВ, затем при нажатии авкОСВ1 - ОСВ по 62 счету не формируется, а вместо этого фокус переходит на форму сформированного перед этим табличного документа ОСВ ?
Как с этим бороться ?

```1c
    &НаСервере
    Функция вчПолучитьСчет62()
    Возврат ПланыСчетов.Хозрасчетный.НайтиПоКоду("62");
    КонецФункции

    // В.Ч. 17.11.2020
    &НаКлиенте
    Процедура авкОСВ1(Команда)
    // ОСВ по счету 62
    авкОСВ(Команда, вчПолучитьСчет62());
    КонецПроцедуры

    //
    &НаКлиенте
    Процедура авкОСВ(Команда, ПоСчету=Неопределено)

    //Имя вызываемого отчета карточка,анализ, ОСВ по счету . Я делал только для этих трех.
    ИмяОтчета = "ОборотноСальдоваяВедомостьПоСчету";

    Группировки = Новый Массив;
    ОписаниеГруппировки = Новый Структура("Использование, Поле", Истина, "Субконто1");
    Группировки.Добавить(ОписаниеГруппировки);
    ОписаниеГруппировки = Новый Структура("Использование, Поле", Истина, "Субконто2");
    Группировки.Добавить(ОписаниеГруппировки);

    //Заполняем реквизиты отчета и обязательные поля (могут отличаться в зависимости от отчета)
    ПользовательскиеНастройкиКомпоновкиДанных = Новый ПользовательскиеНастройкиКомпоновкиДанных;
    Если Объект.РасшифровкаПлатежа.Количество() = 0 Тогда
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = "Не выбран Объект расчетов";
    Сообщение.Сообщить();
    Возврат;
    КонецЕсли;

    // В.Ч. 17.11.2020
    Если ПоСчету <> Неопределено Тогда
    НомерСчета = ПоСчету;
    Иначе
    Если не ЗначениеЗаполнено(Объект.РасшифровкаПлатежа[0].Заказ) Тогда
    НомерСчета = авкПолучитьСчетИзСтатьиРасходов(Объект.РасшифровкаПлатежа[0].СтатьяРасходов);
    Иначе
    НомерСчета = авкПолучитьСчетИзДоговора(Объект.РасшифровкаПлатежа[0].Заказ);
    КонецЕсли;
    КонецЕсли;

    ПериодОтчета = авкПолучитьПериодОтчета(Объект.Дата);

    ПользовательскиеНастройкиКомпоновкиДанных.ДополнительныеСвойства.Вставить("Счет" , НомерСчета); //ПредопределенноеЗначение("ПланСчетов.Хозрасчетный.РасчетыСПоставщикамиИПодрядчиками"));
    ПользовательскиеНастройкиКомпоновкиДанных.ДополнительныеСвойства.Вставить("НачалоПериода" , ПериодОтчета.НачалоПериода);
    ПользовательскиеНастройкиКомпоновкиДанных.ДополнительныеСвойства.Вставить("КонецПериода" , ПериодОтчета.КонецПериода);
    ПользовательскиеНастройкиКомпоновкиДанных.ДополнительныеСвойства.Вставить("Организация" , авкПолучитьОрганизацию());
    ПользовательскиеНастройкиКомпоновкиДанных.ДополнительныеСвойства.Вставить("Группировка" , Группировки);

    //Заполняем дополнительные отборы (в примере только для субконто1 и субконто2)
    НовыйОтбор = ПользовательскиеНастройкиКомпоновкиДанных.Элементы.Добавить(Тип("ОтборКомпоновкиДанных"));
    НовыйОтбор.ИдентификаторПользовательскойНастройки = "Отбор";
    ЗаполнитьЭлементОтбораКомпоновкиДанных(НовыйОтбор,Истина,ВидСравненияКомпоновкиДанных.Равно,"Субконто1", Объект.Контрагент);

    //Обязателньо указываем ключ (Равным параметру "ИДРасшифровки",я указывал Имя отчета ) НастройкиРасшифровки
    НастройкиРасшифровки = Новый Структура;
    НастройкиРасшифровки.Вставить(ИмяОтчета, ПользовательскиеНастройкиКомпоновкиДанных);

    УсловияОтбора = Новый Структура();
    УсловияОтбора.Вставить("НастройкиРасшифровки", НастройкиРасшифровки);

    //Дублируем настройки (у меня в файловой без этого не открывалось)
    ОбщиеНастройки = Новый Структура();
    ОбщиеНастройки.Вставить("Объект" , УсловияОтбора);
    ОбщиеНастройки.Вставить("НастройкиРасшифровки", НастройкиРасшифровки);

    //Помещаем во временное хранилище (Уид можно случайный)
    АдресХранилища = ПоместитьВоВременноеХранилище(ОбщиеНастройки, Новый УникальныйИдентификатор);

    //Обращаем внимания на заполнемые настройки, если вы укажете отбор = истина то затрется все выше указанные отборы ( поскольку показатели и группировки я не менял, то оставил их типоыми)
    ЗаполнятьТиповыеНастройки = Новый Структура;
    ЗаполнятьТиповыеНастройки.Вставить("Отбор" , Ложь);
    ЗаполнятьТиповыеНастройки.Вставить("Группировка" , Ложь);
    ЗаполнятьТиповыеНастройки.Вставить("ВыводимыеДанные", Истина);
    ЗаполнятьТиповыеНастройки.Вставить("Показатели" , Истина);

    //Сами параметры для предачи на форму ("ИДРасшифровки" - в данном случаи ключ от структуры НастройкиРасшифровки , если "СформироватьПриОткрытии" будет равно Ложь то отчет откроется с ранее сохраненными настройками)
    ПараметрыОтчета = Новый Структура;
    ПараметрыОтчета.Вставить("ВидРасшифровки" , 1);
    ПараметрыОтчета.Вставить("АдресНастроек" , АдресХранилища);
    ПараметрыОтчета.Вставить("СформироватьПриОткрытии", Истина);
    ПараметрыОтчета.Вставить("ИДРасшифровки" , ИмяОтчета);
    ПараметрыОтчета.Вставить("РежимРасшифровки" , Истина);
    ПараметрыОтчета.Вставить("ЗаполняемыеНастройки" , ЗаполнятьТиповыеНастройки);

    //Ну и открываем саму форму
    ОткрытьФорму("Отчет." + ИмяОтчета + ".Форма.ФормаОтчета", ПараметрыОтчета, ЭтаФорма, );

    КонецПроцедуры
```
---
При открытии формы нужно указать [Уникальность]
Форма ОткрытьФорму(ИмяФормы, [Параметры], [Владелец], [Уникальность], [Окно], [НавигационнаяСсылка], [ОповещениеОЗакрытии], [РежимОткрытияОкна])

```1c
    //Ну и открываем саму форму // В.Ч. 17.11.2020
    ОткрытьФорму("Отчет." + ИмяОтчета + ".Форма.ФормаОтчета", ПараметрыОтчета, ЭтаФорма, Новый УникальныйИдентификатор());
```
---
При открытии формы нужно указать [Уникальность]
Форма ОткрытьФорму(ИмяФормы, [Параметры], [Владелец], [Уникальность], [Окно], [НавигационнаяСсылка], [ОповещениеОЗакрытии], [РежимОткрытияОкна])

```1c
    //Ну и открываем саму форму // В.Ч. 17.11.2020
    ОткрытьФорму("Отчет." + ИмяОтчета + ".Форма.ФормаОтчета", ПараметрыОтчета, ЭтаФорма, Новый УникальныйИдентификатор());
```
---
Что нам говорит отладчик? Подозреваю, что надо задать уникальность формы. Открыть два одинаковых отчета в разных окнах просто так не получится - только через разделение уникальности.

Кстати, уберите серверную процедуру. Предопределенные значения можно и на клиенте получить.